Busca em Largura – Código fonte
% Predicado que verifica qual é o estado inicial do problema em questão, executa a Busca em Largura e imprime a solução encontrada.

resolva_bl :- estado_inicial(E),

              busca_l([[r(raiz,E)]],Solucao),

              imprima(Solucao), nl.

% Verifica se a cabeça da lista é o estado final.
busca_l([T|_],Solucao):- T atinge_a_meta, !, Solucao = T.

% Encontra todos os filhos do estado que esta na cabeça da lista, os insere no final da lista de estados, e chama a busca em largura recursivamente.
busca_l([T|Fila],Solucao):-

      ache_todos(ExtensaoAteUmFilho,estende_ate_filho(T,ExtensaoAteUmFilho),Extensoes),

      concatena(Fila,Extensoes,FilaEstendida),

      busca_l(FilaEstendida,Solucao).
Busca em profundidade – Código fonte
% Predicado que verifica qual é o estado inicial do problema em questão, executa a Busca em Profundidade e imprime a solução encontrada.
resolva_bp :- estado_inicial(E),

              busca_p([[r(raiz,E)]],Solucao),

              imprima(Solucao), nl.

% Verifica se a cabeça da lista eh o estado final.
busca_p([T|_],Solucao):- T atinge_a_meta, !, Solucao = T.

% Encontra todos os filhos do estado que esta na cabeça da lista, os insere no inicio da lista de estados, e chama a busca em profundidade recursivamente.
busca_p([T|Pilha],Solucao):-

      ache_todos(ExtensaoAteUmFilho,estende_ate_filho(T,ExtensaoAteUmFilho),Extensoes),

      concatena(Extensoes,Pilha,PilhaEstendida),

      busca_p(PilhaEstendida,Solucao).

Predicados Complementares aos dois Tipos de Busca 

% Procura por um filho de um estado, desde que esse filho resulte de uma operação permitida e também não esteja no caminho atual (evitando assim a formação de ciclos).
estende_ate_filho([r(Ramo,N)|Caminho],[r(Op,Filho),r(Ramo,N)|Caminho]):-

                  operacao(Op) transforma N em Filho,

                  Filho nao_produz_circulos_em Caminho.

% Verifica se um dado estado já pertence ao caminho atual.
Estado nao_produz_circulos_em Caminho :-

                  \+ pertence(r(Operacao,Estado),Caminho).

% Verificação do estado final para o problema em questão.
[r(Ramo,M)|_] atinge_a_meta :- M e_a_meta.

Programas Auxiliares
% Predicado que concatena duas listas e coloca o resultado numa terceira lista.
concatena([],X,X) :- !.

concatena([X|Y],Z,[X|W]) :- concatena(Y,Z,W).

% Verifica se um determinado elemento pertence a uma lista.
pertence(X,[X|_]) :- !.

pertence(X,[_|Y]) :- pertence(X,Y).

% Encontra todas as clausulas que satisfaçam a uma dada condição, e as armazena numa lista (nesse caso, Z).
ache_todos(X,Y,Z) :- bagof(X,Y,Z), !.

ache_todos(_,_,[]).

% Imprime o estado inicial, e todos os demais passos e estados até encontrar o estado final.
imprima([r(raiz,Raiz)]) :- !,




   write('Estado inicial: '),




   write(Raiz), 




   write('.'), nl.

imprima([r(Ramo,No)|Resto]) :- imprima(Resto),




       nl,




       write(Ramo),




       write(': '),




       write(No),




       write('.'), nl.

